
Corrigé TP 1 : Algorithme de gradient

Solution du dernier exercice : comparaison de la vitesse de l'algorithme de gradient à pas fixe, à pas optimal et à pas optimal "randomisé" pour la résolution de

$$Ax = b \iff \min_{x \in \mathbb{R}^N} \frac{1}{2} \langle Ax, x \rangle - \langle b, x \rangle.$$

On a choisi

$$A = \text{diag}([1 : 1 : 100]), \quad x_{\text{sol}} = \text{ones}(N, 1) \text{ et } x_0 = \text{rand}(N, 1).$$

```
//Gradient a pas optimal random pour probleme quadratique
clear
clf
N=100;
//A=2*diag(ones(N,1),0)-diag(ones(N-1,1),-1)-diag(ones(N-1,1),1);
A=diag([1:1:N]);
xsol=ones(N,1);
b=A*xsol;
x0=rand(N,1);
itermax=100;
//1.gradient pas fixe
x=x0;
Tfixe=[norm(x-xsol,'inf')];
mu=0.01;
for k=1:itermax
    d=A*x-b;
    x=x-mu*d;
    err=norm(x-xsol,'inf');
    Tfixe=[Tfixe err];
end

//2.gradient pas optimal
x=x0;
Topt=[norm(x-xsol,'inf')];
for k=1:itermax
    d=A*x-b;
    mu=d'*d/(d'*A*d);
    x=x-mu*d;
    err=norm(x-xsol,'inf');
    Topt=[Topt err];
end
```

```

//gradient a pas optimal randomis
x=x0;
Trand=[norm(x-xsol,'inf')];
for k=1:itermax
    d=A*x-b;
    mu=d'*d/(d'*A*d);
    th=2*rand(1,1);
    x=x-th*mu*d;
    err=norm(x-xsol,'inf');
    Trand=[Trand err];
end

plot2d([0:1:itermax]',[Tfixe' Topt' Trand'],logflag='nl',style=[1 2 3]);
xtitle('','k','||xk-x||')
legends(['pas fixe';'pas optimal';'randomisé'],[1 2 3],opt='ll')

```




Figure 1: Vitesse de convergence des algorithmes de gradient